Nyquist–Shannon sampling theorem

Arash Mafi
Assistant Professor
Department of Electrical Engineering and Computer Science
University of Wisconsin-Milwaukee
9/10/2008

Consider the signal x as function of time t in the form of an exponential function $x(t) = \text{Exp}[-t^2]$. Let's plot this function. We see that for |x|> 3, the function is very small.

```
Clear[x, t];
x[t_] = Exp[-t^2];
plot1 = Plot[x[t], {t, -3, 3}, PlotRange → All, PlotStyle → Red, AxesLabel → {"t", "x(t)"}]
```



The Fourier Transform of this function can be easily evaluated and is X(f)=$\frac{e^{-f^2 \pi^2}}{\sqrt{2}}$. If we plot the Fourier transform, we see that it becomes very small for |f|>1. This indicates that the one-sided bandwidth of this function B must be around B=1.0

```
X[f_] = FourierTransform[x[t], t, 2 π f]
Plot[X[f], {f, -1, 1}, PlotRange → All, PlotStyle → Red, AxesLabel → {"f", "X(f)"}]
```

$$\frac{e^{-f^2 \pi^2}}{\sqrt{2}}$$



The Nyquist–Shannon sampling theorem tells us to choose a sampling rate fs at least equal to twice the bandwidth, i.e. fs=2B. The sampled signal is x(nT) for all values of integer n. In practice, a finite number of n is sufficient in this case since x(nT) is vanishingly small for large n. We chose n-nMax=10 for the maximum value of n. Now, from the digital signal x(nT), we try to reconstruct the analog signal y(t) which should ideally be very close to x(t) if the smapling rate is properly chosen. We plot y(t)

```
B = 1;
fs = 2 B;
T = 1/fs;
nMax = 10;
y[t_] = Sum[x[n T] Sinc[π (t - n T)/T], {n, -nMax, nMax}];
plot2 = Plot[y[t], {t, -3, 3}, PlotRange → All,
  PlotStyle → {Dashed, Blue}, AxesLabel → {"t", "y(t)"}]
```

$$y[t\_] = \sum_{n=-nMax}^{nMax} x[n\,T]\ \mathrm{Sinc}\left[\pi \frac{t - n\,T}{T}\right];$$



Let' s compare x (t) in red and y (t) in blue. They are right on top of each other because we chose the sampling rate according to the the sampling theorem.

```
Show[plot1, plot2]
```



Let' s now sample the signal at a lower rate than what the sampling theorem suggests. For example, we choose our bandwidth B too small, say B = 0.5. The original signal x(t) differs from the recostructed signal y(t).

```
B = 0.5;
fs = 2 B;
T = 1/fs;
nMax = 10;
y[t_] = Sum[x[n T] Sinc[π (t - n T)/T], {n, -nMax, nMax}];
plot3 = Plot[y[t], {t, -3, 3}, PlotRange → All,
  PlotStyle → {Dashed, Green}, AxesLabel → {"t", "y(t)"}]
```



```
Show[plot1, plot3]
```



If we lower the sampling rate even more, the disagreement between x (t) and y (t) becomes more apparant.

```
B = 0.2;
fs = 2 B;
T = 1/fs;
nMax = 10;
y[t_] = Sum[x[n T] Sinc[π (t - n T)/T], {n, -nMax, nMax}];
plot4 = Plot[y[t], {t, -3, 3}, PlotRange → All,
  PlotStyle → {Dashed, Purple}, AxesLabel → {"t", "y(t)"}]
```

$$T = \frac{1}{fs};\quad y[t\_] = \sum_{n=-nMax}^{nMax} x[n\,T]\,\mathrm{Sinc}\left[\pi\,\frac{t - n\,T}{T}\right];$$



```
Show[plot1, plot4]
```